Laplace Transform in Maple

Mth 256 March 7 2001

Bent E. Petersen

Filename: 256winter2001_laplace.mws

```
> restart;
> with(inttrans): with(plots):
```

In addition to just computing Laplace and inverse Laplace transforms, Maple can apply the Laplace transform directly to a linear differential equation:

```
> ode1:=diff(y(t),t,t)+4*y(t)=cos(t);
```

$$ode1 := \left(\frac{\partial^2}{\partial t^2}\mathrm{y}(t)\right) + 4\,\mathrm{y}(t) = \cos(t)$$

```
> Lap1:=laplace(ode1,t,s);
```

$$Lap1 := s\,(s\,\mathrm{laplace}(\mathrm{y}(t), t, s) - \mathrm{y}(0)) - \mathrm{D}(y)(0) + 4\,\mathrm{laplace}(\mathrm{y}(t), t, s) = \frac{s}{s^2+1}$$

We can make the equation look comfortably familiar by introducing $\mathrm{Y}(t)$ for the Laplace transform of the unknown function $\mathrm{y}(t)$.

```
> Lp1:=subs(laplace(y(t),t,s)=Y(s),Lap1);
```

$$Lp1 := s\,(s\,\mathrm{Y}(s) - \mathrm{y}(0)) - \mathrm{D}(y)(0) + 4\,\mathrm{Y}(s) = \frac{s}{s^2+1}$$

Now we let Maple do the algebra

```
> solve(Lp1,Y(s)): Y(s):=%;
```

$$\mathrm{Y}(s) := \frac{\mathrm{y}(0)\,s^3 + s\,\mathrm{y}(0) + \mathrm{D}(y)(0)\,s^2 + \mathrm{D}(y)(0) + s}{s^4 + 5\,s^2 + 4}$$

To solve the differential equation we can now take the inverse Laplace transform

```
> soln1:=invlaplace(Y(s),s,t);
```

$$soln1 := -\frac{1}{3}\cos(2\,t) + \mathrm{y}(0)\cos(2\,t) + \frac{1}{2}\mathrm{D}(y)(0)\sin(2\,t) + \frac{1}{3}\cos(t)$$

Notice the solutions are conveniently paramterized by the initial values $y(0)$ and $D(y)(0)$.

Of course, if we just want to solve the differential equation we can do so directly:

```
> init1:=y(0)=A,D(y)(0)=B;
```

$$init1 := y(0) = A, D(y)(0) = B$$

```
> soln1b:=dsolve({ode1,init1},y(t));
```

$$soln1b := y(t) = \left(\frac{1}{4}\sin(t) + \frac{1}{12}\sin(3\,t)\right)\sin(2\,t) + \left(\frac{1}{12}\cos(3\,t) + \frac{1}{4}\cos(t)\right)\cos(2\,t) + \frac{1}{2}B\sin(2\,t) + \left(-\frac{1}{3} + A\right)\cos(2\,t)$$

```
> subs(y(0)=A,D(y)(0)=B,soln1)-rhs(soln1b): simplify(%);
```

$$0$$

In spite of initial appearances the two solutions agree!

The nice thing about the Laplace transform of course is that it correctly handles the case where the driving term has discontinuities. Let's look at a simple example.

```
> ode2:=diff(y(t),t,t)-diff(y(t),t)+6*y(t)=Heaviside(t-1)-Heavisi
de(t-2);
```

$$ode2 := \left(\frac{\partial^2}{\partial t^2}y(t)\right) - \left(\frac{\partial}{\partial t}y(t)\right) + 6\,y(t) = \text{Heaviside}(t-1) - \text{Heaviside}(t-2)$$

```
> Lap2:=laplace(ode2,t,s);
```

$$Lap2 := s\,(s\,\text{laplace}(y(t), t, s) - y(0)) - D(y)(0) - s\,\text{laplace}(y(t), t, s) + y(0) + 6\,\text{laplace}(y(t), t, s) = \frac{e^{(-s)}}{s} - \frac{e^{(-2\,s)}}{s}$$

Before we substitute $Y(s)$ we had best unassign it. Otherwise we will be substituting its previous value rather than just a new symbol for $\text{laplace}(y(t), t, s)$.

```
> unassign('Y(s)');
> Lp2:=subs(laplace(y(t),t,s)=Y(s),Lap2);
```

$$Lp2 := s\,(s\,Y(s) - y(0)) - D(y)(0) - s\,Y(s) + y(0) + 6\,Y(s) = \frac{e^{(-s)}}{s} - \frac{e^{(-2\,s)}}{s}$$

```
> solve(Lp2,Y(s)): Y(s):=%;
```

$$Y(s) := \frac{s^2\, y(0) + D(y)(0)\, s - s\, y(0) + e^{(-s)} - e^{(-2\, s)}}{s\, (s^2 - s + 6)}$$

```
> soln2:=invlaplace(Y(s),s,t);
```

$$soln2 := -\frac{1}{23}\, y(0)\, e^{(1/2\, t)} \sqrt{23}\, \sin\left(\frac{1}{2}\sqrt{23}\, t\right) + y(0)\, e^{(1/2\, t)} \cos\left(\frac{1}{2}\sqrt{23}\, t\right) + \frac{2}{23}\, D(y)(0)\, e^{(1/2\, t)} \sqrt{23}\, \sin\left(\frac{1}{2}\sqrt{23}\, t\right) + \frac{1}{6}\, \mathrm{Heaviside}(t-1) + \frac{1}{138}\, \mathrm{Heaviside}(t-1)\, e^{(1/2\, t - 1/2)} \sqrt{23}\, \sin\left(\frac{1}{2}\sqrt{23}\, (t-1)\right) - \frac{1}{6}\, \mathrm{Heaviside}(t-1)\, e^{(1/2\, t - 1/2)} \cos\left(\frac{1}{2}\sqrt{23}\, (t-1)\right) - \frac{1}{6}\, \mathrm{Heaviside}(t-2) - \frac{1}{138}\, \mathrm{Heaviside}(t-2)\, e^{(1/2\, t - 1)} \sqrt{23}\, \sin\left(\frac{1}{2}\sqrt{23}\, (t-2)\right) + \frac{1}{6}\, \mathrm{Heaviside}(t-2)\, e^{(1/2\, t - 1)} \cos\left(\frac{1}{2}\sqrt{23}\, (t-2)\right)$$

```
> exmp2:=subs(y(0)=0,D(y)(0)=0,soln2);
```

$$exmp2 := \frac{1}{6}\, \mathrm{Heaviside}(t-1) + \frac{1}{138}\, \mathrm{Heaviside}(t-1)\, e^{(1/2\, t - 1/2)} \sqrt{23}\, \sin\left(\frac{1}{2}\sqrt{23}\, (t-1)\right) - \frac{1}{6}\, \mathrm{Heaviside}(t-1)\, e^{(1/2\, t - 1/2)} \cos\left(\frac{1}{2}\sqrt{23}\, (t-1)\right) - \frac{1}{6}\, \mathrm{Heaviside}(t-2) - \frac{1}{138}\, \mathrm{Heaviside}(t-2)\, e^{(1/2\, t - 1)} \sqrt{23}\, \sin\left(\frac{1}{2}\sqrt{23}\, (t-2)\right) + \frac{1}{6}\, \mathrm{Heaviside}(t-2)\, e^{(1/2\, t - 1)} \cos\left(\frac{1}{2}\sqrt{23}\, (t-2)\right)$$

```
> plot(exmp2,t=0..3,thickness=3,color=red);
```



Be sure to check the Maple help facility for more information and examples.

Note as mentioned at the beginning Maple can also be used just to compute transforms. Here's a couple of examples:

```
> laplace(exp(at)*sin(omega*t-phi),t,s);
```

$$-e^{at}\left(-\frac{\cos(\phi)\,\omega}{s^2+\omega^2}+\frac{\sin(\phi)\,s}{s^2+\omega^2}\right)$$

```
> invlaplace((2*s-3)/((s^2+4)^2*(s+1)),s,t);
```

$$-\frac{1}{5}e^{(-t)}+\frac{1}{5}\cos(2\,t)-\frac{3}{80}\sin(2\,t)-\frac{1}{8}t\cos(2\,t)+\frac{1}{4}t\sin(2\,t)$$

That sure beats doing the algebra by hand!

```
>
```